
The Fractal Dimension Index is a technical indicator that gauges the amount of volatility currently present in the market.

The theory behind this indicator is that a value of 1.5 suggests the market is acting in a completely random fashion. As the market deviates from 1.5, the opportunity for earning profits is increased
in proportion to the amount of deviation.

Keep in mind that the indicator does not show the direction of trends! Although you can try to test it as a trend-following indicator that gives trend-following signals, that isn't the intended use of
the indicator.

The Fractal Dimension Index is red when the market is in a trend. And it is blue when there is high volatility. When the Fractal Dimension Index changes its color from red to blue, it means that a
trend is finishing. The market becomes erratic and high in volatility when the Fractal Dimension Index is blue. Usually, these "blue periods" do not go on for a long time, they come before a new
trend.

So, look for trend-following signals while the Fractal Dimension Index is blue since this indicates high volatility before a potential trend, and avoid trend-following signals when the Fractal
Dimension Index is red since this indicates a ranging/non-trending market or a trend that started long ago.

Open-source script
In true TradingView spirit, the author of this script has published it open-source, so traders can understand and verify it. Cheers to the author! You may use it for free, but reuse of this code in a publication is governed
by House Rules. You can favorite it to use it on a chart.

Disclaimer
The information and publications are not meant to be, and do not constitute, financial, investment, trading, or other types of advice or recommendations supplied or endorsed by TradingView. Read more in the Terms
of Use.

Want to use this script on a chart? 

 Add to favorite indicators

// This source code is subject to the terms of the Mozilla Public License 2.0 at https://mozilla.org/MPL/2.0/
// © bjr117

//@version=5
indicator(title = "Fractal Dimension Index", shorttitle = "FDI", overlay = false)

//==============================================================================
// Input Parameters
//==============================================================================
fdi_length = input.int(30, title = "FDI Length", minval = 2)
fdi_src    = input.source(close, title = "FDI Source")
fdi_hline  = input.float(1.5, title = "FDI Horizontal Line Level", minval = 0.0, maxval = 2.0, step = 0.005)
//==============================================================================

//==============================================================================
// Calculating FDI
//==============================================================================
calculate_FDI(fdi_src, fdi_length) =>
    highest_high = ta.highest(fdi_src, fdi_length)
    lowest_low   = ta.lowest(fdi_src,  fdi_length)

    length = float(0.0)
    for i = 1 to fdi_length - 1
        diff = (fdi_src[i] - lowest_low) / (highest_high - lowest_low)
        length := length + math.sqrt(math.pow(diff[i] - diff[i+1], 2) + (1 / math.pow(fdi_length, 2)))

    fdi = 1 + (math.log(length) + math.log(2)) / math.log(2*fdi_length)
    fdi
//==============================================================================

//==============================================================================
// Plotting FDI and FDI Horizontal Line
//==============================================================================
fdi  = float(0.0)
fdi := calculate_FDI(fdi_src, fdi_length)

fdi_color = fdi > fdi_hline ?  color.blue :  color.red
fdi_plot   = plot(fdi, title = "FDI Line", color = fdi_color)
hline_plot = plot(fdi_hline, title = "FDI Horizontal Line", color =  color.new(#000000, 70))
fill(fdi_plot, hline_plot, title = "FDI Fill", color = color.new(fdi_color, 70))
//==============================================================================

